

Solution of ODE's and eigenvalue problems with a Chebyshev polynomial spectral method

This notebook has been written in *Mathematica* by

Mark J. McCready
Professor and Chair of Chemical Engineering
University of Notre Dame
Notre Dame IN 46556
USA

Mark.J.McCready.1@nd.edu
<http://www.nd.edu/~mjm/>

It is copyrighted to the extent allowed by whatever laws pertain to the World Wide Web and the Internet.

I would hope that as a professional courtesy, that if you use it, that this notice remain visible to other users.
There is no charge for copying and dissemination

Version: 11/28/98

This notebook demonstrates the Orszag-tau spectral method for a simple ODE and an ODE eigenvalue problem. It is intended as a first introduction to solving these simple problems with a spectral method.

Reference: S. A. Orszag (1971) "Accurate solution of the Orr-Sommerfeld stability equation", *Journal of Fluid Mechanics*, **50** pp 689-703.

The coefficients of the algebraic equations are computed directly from the orthogonality property of the Chebyshev polynomials using integration.

Reference: R. Miesen and B. J. Boersma (1995) "Hydrodynamic stability of a sheared liquid film", *Journal of Fluid Mechanics*, **301** pp 175-202.

Mathematica aside

Some properties of Chebyshev polynomials

For this example, we will be using the ChebyshevT polynomials which are orthogonal and non singular on the interval $[-1,1]$. They also have the value of $T[i,1]=T[i,-1]=\pm 1$. Let's see what these are


```
somechebys = Table[ChebyshevT[i, y], {i, 0, 10}]
```

```
{1, y, 2 y^2 - 1, 4 y^3 - 3 y, 8 y^4 - 8 y^2 + 1, 16 y^5 - 20 y^3 + 5 y,
 32 y^6 - 48 y^4 + 18 y^2 - 1, 64 y^7 - 112 y^5 + 56 y^3 - 7 y, 128 y^8 - 256 y^6 + 160 y^4 - 32 y^2 + 1,
 256 y^9 - 576 y^7 + 432 y^5 - 120 y^3 + 9 y, 512 y^10 - 1280 y^8 + 1120 y^6 - 400 y^4 + 50 y^2 - 1}
```

If we wish, we can plot all of these.

The Evaluate is needed to make the function be recognized before it is numerically evaluated.

```
Plot[Evaluate[somechebys], {y, -1, 1}]
```



- Graphics -

This is too complicated to see much. But there are even polynomials and odd ones with the expected symmetry. Also, it can be seen that the interval is spanned effectively with these so they look to be good expansion functions.

These are orthogonal, let's see how

```
ChebyshevT[0, y]
```

What about some derivatives.

D[ChebyshevT[i, y], {y, 1}]

$$i U_{i-1}(y)$$

D[ChebyshevT[i, y], {y, 2}]

$$\frac{i(i T_i(y) - y U_{i-1}(y))}{y^2 - 1}$$

D[ChebyshevT[i, y], {y, 3}]

$$\frac{i(U_{i-1}(y) i^2 - U_{i-1}(y) - \frac{y(i T_i(y) - y U_{i-1}(y))}{y^2 - 1})}{y^2 - 1} - \frac{2 i y (i T_i(y) - y U_{i-1}(y))}{(y^2 - 1)^2}$$

D[ChebyshevT[i, y], {y, 4}]

$$\begin{aligned} & \frac{8 i (i T_i(y) - y U_{i-1}(y)) y^2}{(y^2 - 1)^3} - \\ & \frac{4 i (U_{i-1}(y) i^2 - U_{i-1}(y) - \frac{y(i T_i(y) - y U_{i-1}(y))}{y^2 - 1}) y}{(y^2 - 1)^2} - \frac{2 i (i T_i(y) - y U_{i-1}(y))}{(y^2 - 1)^2} + \\ & \frac{1}{y^2 - 1} \left(i \left(\frac{(i T_i(y) - y U_{i-1}(y)) i^2}{y^2 - 1} - \frac{2 (i T_i(y) - y U_{i-1}(y))}{y^2 - 1} + \frac{2 y^2 (i T_i(y) - y U_{i-1}(y))}{(y^2 - 1)^2} - \right. \right. \\ & \left. \left. \frac{y (U_{i-1}(y) i^2 - U_{i-1}(y) - \frac{y(i T_i(y) - y U_{i-1}(y))}{y^2 - 1})}{y^2 - 1} \right) \right) \end{aligned}$$

Basic idea of how to use a spectral numerical method to solve an ODE.

We have seen that we can fit an arbitrary function using a set of orthogonal polynomials by forming the inner product to evaluate the coefficients. We thus *conjecture* that we could solve an ode, which must have some function form (albeit) unknown, with an appropriate set of orthogonal polynomials, or other orthogonal functions. The obvious questions are which functions and how do we get the coefficients.

The starting point is an assumed solution form of:

$$u[x] = \sum_{j=1}^{\infty} c_j \phi_j(x)$$

We need the ϕ_j 's to be orthogonal on the domain of interest and capable of producing a boundary conditions of the needed form.

We will substitute the assumed form for the solution into the ODE and boundary conditions, form an inner product with every term (not the boundary conditions however) (why not), and then see if we can evaluate the coefficients. If we can, we have a solution.

Note however, that we will expect to use a finite number of terms -- as opposed to the *series* solution of an ode where we expect to find a recursion relation valid for all terms. We then hope that the solution will get more accurate as we increase the number of terms.

We first start with a very simple ode that has an analytical solution with which to compare.

Solution of a simple ode

Here is a simple ODE that has an easy analytical solution. We will then use a spectral technique and then compare the answers.

$$\frac{d^2 u[y]}{dy^2} + u[y] = 0,$$

$$u[1] = u[-1] = 1$$

■ Analytical solution

$$\text{ode} = \frac{\partial^2 u(y)}{\partial y^2} + u(y)$$

$$u(y) + u''(y)$$


```
ans1 = DSolve[{ode == 0, u[1] == 1, u[-1] == 1}, u[y], y]
```

$$\{\{u(y) \rightarrow \cos(y) \sec(1)\}\}$$

```
soln1 = Simplify[u[y] /. ans1[[1]]]
```

$$\cos(y) \sec(1)$$

```
p1 = Plot[soln1, {y, -1, 1}]
```



- Graphics -

■ Numerical solution

Now solve the equation with a Galerkin (name of the person associated with this kind of spectral approach) spectral technique that involves substituting a series of Chebyshevs into the differential equation, employing orthogonality and then solving the algebraic equations. We use direct substitution and don't attempt to reduce the derivatives with analytical manipulations.

$$\text{ode1} = \frac{\partial^2 \text{vv}(y)}{\partial y^2} + \text{vv}(y)$$

$$\text{vv}(y) + \text{vv}''(y)$$

The basic assumption of a Galerkin spectral method is that the numerical solution, can be written as a sum of orthogonal functions. This trial solution has a finite number of terms and is valid over the entire domain. If the functions are chosen to automatically fit the boundary conditions (e.g., perhaps by some clever choices of sums of groups of polynomials, or sheer luck).

If the functions do not assure automatic fitting of the boundary conditions, we will have to make some adjustments. The *tau* method, used below, replaces one of the equations for $a[i]$ generated by the ODE, for *each* algebraic equation that is needed to assure boundary condition fitting. Stated again, if we have 4 boundary conditions, we will replace 4 of the equations generated from the ode with the 4 equations that must be fit on the boundary. To minimize the error, we think of this as starting with more terms in the sum than we need, then dropping of the highest "frequency" modes that are assumed (this is confirmed) to be less important to getting an accurate solution than the lowest modes.

Here is where we substitute Chebyshev polynomials for $\text{vv}[y]$.

```
temp1 = ode1 /. {vv[y] -> a[i] ChebyshevT[i, y],
  D_{y,a1} vv[y] -> D_{y,a1} (a[i] ChebyshevT[i, y])}
```

$$a(i) T_i(y) + \frac{i a(i) (i T_i(y) - y U_{i-1}(y))}{y^2 - 1}$$

```
temp2 = ExpandAll[temp1]
```

$$\frac{a(i) T_i(y) i^2}{y^2 - 1} - \frac{y a(i) U_{i-1}(y) i}{y^2 - 1} + a(i) T_i(y)$$

Now get a finite size sum to solve.

n = 6 ;

Because of the symmetry of the boundary conditions and the occurrence of only even derivatives, the solution will be an even function so we need only even polynomials. The $\Delta i=2$, means to skip every other term.

$$\mathbf{temp3} = \sum_{\substack{i=0 \\ \Delta i=2}}^n \mathbf{temp2}$$

$$\begin{aligned} & -\frac{4a(2)y^2}{y^2-1} - \frac{4(8y^3-4y)a(4)y}{y^2-1} - \frac{6(32y^5-32y^3+6y)a(6)y}{y^2-1} + a(0) + \\ & \frac{4(2y^2-1)a(2)}{y^2-1} + (2y^2-1)a(2) + \frac{16(8y^4-8y^2+1)a(4)}{y^2-1} + (8y^4-8y^2+1)a(4) + \\ & \frac{36(32y^6-48y^4+18y^2-1)a(6)}{y^2-1} + (32y^6-48y^4+18y^2-1)a(6) \end{aligned}$$

This might go quicker if we simplify it.

temp4 = Simplify[temp3]

$$32a(6)y^6 + 8a(4)y^4 + 912a(6)y^4 + 88a(4)y^2 - 558a(6)y^2 + a(0) + (2y^2+3)a(2) - 15a(4) + 35a(6)$$

We need to find algebraic equations for the $a[i]$'s. We will use the orthogonality property of the polynomials to pick out equations for each $a[i]$.

$$\int_{-1}^1 \rho(y) * T_j(y) * (\text{substituted ODE with } a \text{ number of } T_i(y)'s) dy .$$

One other "trick" needs to be recognized. The orthogonality property will take care of "choosing" the nonzero pieces of the equation for each coefficient even if we multiply everything out ahead of time as we have done above.

Now set up integrals noting the complete tricks needed for orthogonality. We multiply by the weighting function and an arbitrary ChebyshevT. We will let the "i" take on all possible values between 0 and n to form n/2 equations for the n/2 coefficients. We note that the integration will take of itself

$$\text{temp5} = \frac{\text{temp4} \frac{1}{\sqrt{1-y^2}} \text{ChebyshevT}[j, y] 2}{\pi c[j]}$$

$$\frac{1}{\pi \sqrt{1-y^2} c(j)} (2 (32 a(6) y^6 + 8 a(4) y^4 + 912 a(6) y^4 + 88 a(4) y^2 - 558 a(6) y^2 + a(0) + (2 y^2 + 3) a(2) - 15 a(4) + 35 a(6)) T_j(y))$$

Here we generate the $n/2$ equations in a table form. The original equation was equal to 0, now each of these pieces is also equal to 0.

$$\text{temp6} = \text{Table}[\text{temp5}, \{j, 0, n, 2\}]$$

$$\left\{ \begin{aligned} & \frac{1}{\pi \sqrt{1-y^2} c(0)} (2 (32 a(6) y^6 + 8 a(4) y^4 + 912 a(6) y^4 + 88 a(4) y^2 - 558 a(6) y^2 + a(0) + (2 y^2 + 3) a(2) - 15 a(4) + 35 a(6))), \\ & \frac{1}{\pi \sqrt{1-y^2} c(2)} (2 (2 y^2 - 1) (32 a(6) y^6 + 8 a(4) y^4 + 912 a(6) y^4 + 88 a(4) y^2 - 558 a(6) y^2 + a(0) + (2 y^2 + 3) a(2) - 15 a(4) + 35 a(6))), \\ & \frac{1}{\pi \sqrt{1-y^2} c(4)} (2 (8 y^4 - 8 y^2 + 1) (32 a(6) y^6 + 8 a(4) y^4 + 912 a(6) y^4 + 88 a(4) y^2 - 558 a(6) y^2 + a(0) + (2 y^2 + 3) a(2) - 15 a(4) + 35 a(6))), \\ & \frac{1}{\pi \sqrt{1-y^2} c(6)} (2 (32 y^6 - 48 y^4 + 18 y^2 - 1) (32 a(6) y^6 + 8 a(4) y^4 + 912 a(6) y^4 + 88 a(4) y^2 - 558 a(6) y^2 + a(0) + (2 y^2 + 3) a(2) - 15 a(4) + 35 a(6))) \end{aligned} \right\}$$

Let's get rid of the $c[j]$'s

```
temp7=(temp6/.c[0]->2)/.Table[c[j]->1,{i,2,n,2}]
```

$$\left\{ \frac{1}{\pi \sqrt{1-y^2}} (32 a(6) y^6 + 8 a(4) y^4 + 912 a(6) y^4 + 88 a(4) y^2 - 558 a(6) y^2 + a(0) + (2 y^2 + 3) a(2) - 15 a(4) + 35 a(6)), \right. \\ \frac{1}{\pi \sqrt{1-y^2} c(2)} (2 (2 y^2 - 1) (32 a(6) y^6 + 8 a(4) y^4 + 912 a(6) y^4 + 88 a(4) y^2 - 558 a(6) y^2 + a(0) + (2 y^2 + 3) a(2) - 15 a(4) + 35 a(6))), \\ \frac{1}{\pi \sqrt{1-y^2} c(4)} (2 (8 y^4 - 8 y^2 + 1) (32 a(6) y^6 + 8 a(4) y^4 + 912 a(6) y^4 + 88 a(4) y^2 - 558 a(6) y^2 + a(0) + (2 y^2 + 3) a(2) - 15 a(4) + 35 a(6))), \\ \left. \frac{1}{\pi \sqrt{1-y^2} c(6)} (2 (32 y^6 - 48 y^4 + 18 y^2 - 1) (32 a(6) y^6 + 8 a(4) y^4 + 912 a(6) y^4 + 88 a(4) y^2 - 558 a(6) y^2 + a(0) + (2 y^2 + 3) a(2) - 15 a(4) + 35 a(6))) \right\}$$

Now we complete the calculation by integrating "temp7" from -1 to 1.

```
temp8 = Integrate [temp7, {y, -1, 1}]
```

$$\left\{ a(0) + 4 (a(2) + 8 a(4) + 27 a(6)), \frac{a(2) + 48 (a(4) + 4 a(6))}{c(2)}, \frac{a(4) + 120 a(6)}{c(4)}, \frac{a(6)}{c(6)} \right\}$$

Each of the entries in this table is an equation that equals 0. We can solve this at this point at get the trivial solution since there are 4 unknown a(j)'s and 4 equations. Of course, since we have thus far just substituted into a homogeneous equation and not yet used the nonhomogeneous boundary conditions, what other answer could we expect!

So we need to enforce the nonhomogeneous boundary conditions both to get a nonzero answer and also to get the correct answer.

As mentioned above, we will replace one of these n/2+1 equations with the boundary condition. You might ask which one should we replace. The obvious one(s) to replace are the equations generated by the highest values of i. We justify this two ways. One is that we employ extra terms, which are necessarily at high values of "i" in the expansion so that there are enough to get a good answer. Thus dropping a few at high i does not seem to hurt. A second way to think of this is that the functions have an intrinsic "frequency" and we are just dropping off some of the higher frequencies that are not important.

- **What is the boundary condition in terms of the Chebyshev's?**

We need $u[1]=u[-1]=1$

$$\text{bc1} = \sum_{\substack{i=0 \\ \Delta i=2}}^n a[i]$$

$$a(0) + a(2) + a(4) + a(6)$$

So we will have $n/2$ equations that $=0$ and one (bc1), that equals 1 in our set to solve.

Here is where we take the first $n/2$ rows of the coefficients from the ODE and Append the one boundary condition.

$$\text{temp9} = \text{Append}[\text{Table}[\text{temp8}[i] == 0, \{i, 1, \frac{n}{2}\}], \text{bc1} == 1]$$

$$\left\{ \begin{aligned} a(0) + 4(a(2) + 8a(4) + 27a(6)) == 0, & \frac{a(2) + 48(a(4) + 4a(6))}{c(2)} == 0, & \frac{a(4) + 120a(6)}{c(4)} == 0, \\ a(0) + a(2) + a(4) + a(6) == 1 \end{aligned} \right\}$$

We solve for the coefficients.

$$\text{coef} = \text{Solve}[\text{temp9}, \text{Table}[a[i], \{i, 0, n, 2\}]]$$

$$\left\{ \left\{ a(0) \rightarrow \frac{18540}{13091}, a(2) \rightarrow -\frac{5568}{13091}, a(4) \rightarrow \frac{120}{13091}, a(6) \rightarrow -\frac{1}{13091} \right\} \right\}$$

We want to reconstruct the solution. First make the series.

$$\text{expand1} = \sum_{\substack{i=0 \\ \Delta i=2}}^n a[i] \text{ChebyshevT}[i, y]$$

$$a(0) + (2y^2 - 1)a(2) + (8y^4 - 8y^2 + 1)a(4) + (32y^6 - 48y^4 + 18y^2 - 1)a(6)$$


Then substitute the coefficients that we calculated above to get the answer.

```
anscheby = expand1 /. coef[[1]]
```

$$-\frac{5568(2y^2 - 1)}{13091} + \frac{120(8y^4 - 8y^2 + 1)}{13091} + \frac{-32y^6 + 48y^4 - 18y^2 + 1}{13091} + \frac{18540}{13091}$$

Here is a plot of the numerical solution.


```
p2 = Plot[anscheby, {y, -1, 1}]
```



- Graphics -

Check with the exact solution, we can't see any difference.

```
Show[p1, p2]
```



- Graphics -

Look at the series expansions. First the exact solution:

```
analytseries = Chop[N[Series[soln1, {y, 0, n}]]]
```

$$1.8508 - 0.925408 y^2 + 0.077117 y^4 - 0.00257058 y^6 + O(y^7)$$

Here is the numerical solution:

```
numseries = N[Series[anscheby, {y, 0, n}]]
```

$$1.8508 - 0.925369 y^2 + 0.076999 y^4 - 0.00244443 y^6 + O(y^7)$$

Here is the difference.

```
comp1 = analytseries - numseries
```

$$2.18166 \times 10^{-6} - 0.000039285 y^2 + 0.000117856 y^4 - 0.00012615 y^6 + O(y^7)$$

Here is an integral measure of the error. If we use enough terms, it is difficult for *Mathematica* to compute the error.

```
e[n] = NIntegrate[(soln1 - anscheby), {y, -1, 1}]
```

```
-6.18436 × 10-7
```

This approach will work for any n , but will get real slow as we increase n . Thus the next section employs the same technique but with a little more simplifications before doing the integrals. This will allow reasonable processing times for larger n 's.

Solution of the same simple ode using a slightly more sophisticated approach

- Spectral numerical solution
- Evaluation of convergence
- Numerical solution using just a power series
- Here are some comparison of error for the Chebyshev and power series solutions.

Eigenvalue problem

We will solve the ODE, with an explicit parameter, λ , for homogeneous boundary conditions.

$$u''[y] + (1 + \lambda) u[y] = 0, \quad u[1] = 0, \quad u[-1] = 0$$

- Analytical solution to eigenvalue problem

- Here we get the answer analytically.

When we solve this in *Mathematica*, we don't evaluate both boundary conditions or it will return the homogeneous solution.

```
DSolve[{u''[y] + (1 + λ) u[y] == 0, u[1] == 0}, u[y], y]
```

```
{{u(y) -> e^{-y\sqrt{-λ-1}} (e^{2y\sqrt{-λ-1}} c_2 - e^{2\sqrt{-λ-1}} c_2)}}
```

We can find a simpler even function answer:


```
anseig = A Cos[√(1 + λ) y]
```

```
A cos(y √(λ + 1))
```

- Check this answer


- We can see the eigenvalues from a plot as these are the 0's of the function, $A \cos[\sqrt{1 + \lambda} y]$.

```
Plot[anseig /. {A -> 1, y -> 1}, {λ, 0, 30}]
```



- Graphics -

```
Plot[anseig /. {A → 1, y → 1}, {λ, 0, 240}]
```



- Graphics -

- The roots are easy to figure out:

```
FindRoot[(anseig /. {A → 1, y → 1}) == 0, {λ, 2}]
```

```
{λ → 1.4674}
```

We should be able to figure out the roots, as it is a simple equation!!

$$\sqrt{2.4674}$$

```
1.5708
```

$$N\left[\frac{\pi}{2}\right]$$

```
1.5708
```

```
FindRoot[(anseig /. {A → 1, y → 1}) == 0, {λ, 20}]
```

```
{λ → 21.2066}
```

- Here are some eigenvalues

```
eigsanlyt = Table[N[((i*Pi)/2)^2 - 1], {i, 1, 13, 2}]
```

```
{1.4674, 21.2066, 60.685, 119.903, 198.859, 297.556, 415.991}
```


- Here are some plots of eigenfunctions

$\lambda = 1.4674$

```
afunc1=(anseig/.A->1)/.λ->eigsanlyt[[1]]
```

```
cos(1.5708 y)
```

```
aplot1=Plot[afunc1/.A->1,{y,-1,1}]
```




- Graphics -

$\lambda = 21.2066$

```
afunc2=(anseig/.A->1)/.λ->eigsanlyt[[2]]
```

```
cos(4.71239 y)
```

```
aplot2=Plot[afunc2/.A->1,{y,-1,1}]
```




- Graphics -

```
 $\lambda = 60.7158$ 
```

```
afunc3=(anseig/.A->1)/.λ->eigsanlyt[[3]]
```

```
cos(7.85398 y)
```

```
aplot3=Plot[afunc3/.A->1,{y,-1,1}]
```



- Graphics -

$\lambda = 119.903$

```
afunc4=(anseig/.A->1)/.λ->eigsanlyt[[4]]
```

- Here is the series expansion of the answer
- **Numerical Solution to the eigenvalue problem**
- Numerical solution

Here is our equation again.

$$\text{odeeig1} = \frac{\partial^2 vv(y)}{\partial y^2} + (\lambda + 1) vv(y)$$

$$(\lambda + 1) vv(y) + vv''(y)$$

Here is the standard substitution.

```
t1 = odeeig1 /.
  {vv[y] -> a[i]*ChebyshevT[i, y],
   D[vv[y], {y, a1_}] -> D[a[i]*ChebyshevT[i, y], {y, a1}]}
```

$$(\lambda + 1) a(i) T_i(y) + \frac{i a(i) (i T_i(y) - y U_{i-1}(y))}{y^2 - 1}$$

```
t2 = ExpandAll[t1]
```

$$\frac{a(i) T_i(y) i^2}{y^2 - 1} - \frac{y a(i) U_{i-1}(y) i}{y^2 - 1} + \lambda a(i) T_i(y) + a(i) T_i(y)$$

Now get a finite size sum to solve. Note that we will need only even terms because the solution is clearly even.

```
n = 12;
```

$$t3 = \sum_{\substack{i=0 \\ \Delta i=2}}^n t2$$

This might go quicker if we simplify it.

```
t4 = Simplify[t3]
```

$$\begin{aligned} &2048 \lambda a(12) y^{12} + 2048 a(12) y^{12} + 512 \lambda a(10) y^{10} + 512 a(10) y^{10} - 6144 \lambda a(12) y^{10} + \\ &264192 a(12) y^{10} + 128 \lambda a(8) y^8 + 128 a(8) y^8 - 1280 \lambda a(10) y^8 + 44800 a(10) y^8 + \\ &6912 \lambda a(12) y^8 - 546048 a(12) y^8 + 32 \lambda a(6) y^6 + 32 a(6) y^6 - 256 \lambda a(8) y^6 + 6912 a(8) y^6 + \\ &1120 \lambda a(10) y^6 - 70560 a(10) y^6 - 3584 \lambda a(12) y^6 + 383488 a(12) y^6 + 8 \lambda a(4) y^4 + \\ &8 a(4) y^4 - 48 \lambda a(6) y^4 + 912 a(6) y^4 + 160 \lambda a(8) y^4 - 7520 a(8) y^4 - 400 \lambda a(10) y^4 + \\ &33200 a(10) y^4 + 840 \lambda a(12) y^4 - 106680 a(12) y^4 - 8 \lambda a(4) y^2 + 88 a(4) y^2 + \\ &18 \lambda a(6) y^2 - 558 a(6) y^2 - 32 \lambda a(8) y^2 + 1888 a(8) y^2 + 50 \lambda a(10) y^2 - 4750 a(10) y^2 - \\ &72 \lambda a(12) y^2 + 10008 a(12) y^2 + (\lambda + 1) a(0) + (2 \lambda y^2 + 2 y^2 - \lambda + 3) a(2) + \lambda a(4) - \\ &15 a(4) - \lambda a(6) + 35 a(6) + \lambda a(8) - 63 a(8) - \lambda a(10) + 99 a(10) + \lambda a(12) - 143 a(12) \end{aligned}$$

We need to find equations for the $a[i]$'s. We can use the orthonormality property of the polynomials to pick out equations for each $a[i]$.

Set up integrals

$$t5 = \frac{t4 \frac{1}{\sqrt{1-y^2}} \text{ChebyshevT}[i, y] 2}{\pi c[i]}$$

$$\frac{1}{\pi \sqrt{1-y^2} c(i)} (2 (2048 \lambda a(12) y^{12} + 2048 a(12) y^{12} + 512 \lambda a(10) y^{10} + 512 a(10) y^{10} - 6144 \lambda a(12) y^{10} + 264192 a(12) y^{10} + 128 \lambda a(8) y^8 + 128 a(8) y^8 - 1280 \lambda a(10) y^8 + 44800 a(10) y^8 + 6912 \lambda a(12) y^8 - 546048 a(12) y^8 + 32 \lambda a(6) y^6 + 32 a(6) y^6 - 256 \lambda a(8) y^6 + 6912 a(8) y^6 + 1120 \lambda a(10) y^6 - 70560 a(10) y^6 - 3584 \lambda a(12) y^6 + 383488 a(12) y^6 + 8 \lambda a(4) y^4 + 8 a(4) y^4 - 48 \lambda a(6) y^4 + 912 a(6) y^4 + 160 \lambda a(8) y^4 - 7520 a(8) y^4 - 400 \lambda a(10) y^4 + 33200 a(10) y^4 + 840 \lambda a(12) y^4 - 106680 a(12) y^4 - 8 \lambda a(4) y^2 + 88 a(4) y^2 + 18 \lambda a(6) y^2 - 558 a(6) y^2 - 32 \lambda a(8) y^2 + 1888 a(8) y^2 + 50 \lambda a(10) y^2 - 4750 a(10) y^2 - 72 \lambda a(12) y^2 + 10008 a(12) y^2 + (\lambda + 1) a(0) + (2 \lambda y^2 + 2 y^2 - \lambda + 3) a(2) + \lambda a(4) - 15 a(4) - \lambda a(6) + 35 a(6) + \lambda a(8) - 63 a(8) - \lambda a(10) + 99 a(10) + \lambda a(12) - 143 a(12)) T_i(y))$$

t6 = Table[t5, {i, 0, n, 2}]

t7 = (t6 /. c[0] → 2) /. Table[c[i] → 1, {i, 2, n}]

t8 = Collect[Expand[Numerator[t7]], y]

t9 = Transpose[Table[Coefficient[t8, y, i], {i, 0, 2 n, 2}]]

$$\left(\begin{array}{l} \lambda a(0) + a(0) - \lambda a(2) + 3 a(2) + \lambda a(4) - 15 a(4) - \lambda a(6) + 35 a(6) + \lambda a(8) - 63 a(8) - \lambda a(10) + 99 a(10) - \\ - 2 \lambda a(0) - 2 a(0) + 2 \lambda a(2) - 6 a(2) - 2 \lambda a(4) + 30 a(4) + 2 \lambda a(6) - 70 a(6) - 2 \lambda a(8) + 126 a(8) + 2 \lambda a(10) - \\ - 2 \lambda a(0) + 2 a(0) - 2 \lambda a(2) + 6 a(2) + 2 \lambda a(4) - 30 a(4) - 2 \lambda a(6) + 70 a(6) + 2 \lambda a(8) - 126 a(8) - 2 \lambda a(10) + \\ - 2 \lambda a(0) - 2 a(0) + 2 \lambda a(2) - 6 a(2) - 2 \lambda a(4) + 30 a(4) + 2 \lambda a(6) - 70 a(6) - 2 \lambda a(8) + 126 a(8) + 2 \lambda a(10) - \\ - 2 \lambda a(0) + 2 a(0) - 2 \lambda a(2) + 6 a(2) + 2 \lambda a(4) - 30 a(4) - 2 \lambda a(6) + 70 a(6) + 2 \lambda a(8) - 126 a(8) - 2 \lambda a(10) + \\ - 2 \lambda a(0) - 2 a(0) + 2 \lambda a(2) - 6 a(2) - 2 \lambda a(4) + 30 a(4) + 2 \lambda a(6) - 70 a(6) - 2 \lambda a(8) + 126 a(8) + 2 \lambda a(10) - \\ 2 \lambda a(0) + 2 a(0) - 2 \lambda a(2) + 6 a(2) + 2 \lambda a(4) - 30 a(4) - 2 \lambda a(6) + 70 a(6) + 2 \lambda a(8) - 126 a(8) - 2 \lambda a(10) + \end{array} \right)$$

$$\mathbf{t10} = \text{Table}\left[\frac{y^i}{\pi \sqrt{1-y^2}}, \{i, 0, 2n, 2\}\right]$$

$$\left\{\frac{1}{\pi \sqrt{1-y^2}}, \frac{y^2}{\pi \sqrt{1-y^2}}, \frac{y^4}{\pi \sqrt{1-y^2}}, \frac{y^6}{\pi \sqrt{1-y^2}}, \frac{y^8}{\pi \sqrt{1-y^2}}, \frac{y^{10}}{\pi \sqrt{1-y^2}}, \frac{y^{12}}{\pi \sqrt{1-y^2}}, \frac{y^{14}}{\pi \sqrt{1-y^2}}, \frac{y^{16}}{\pi \sqrt{1-y^2}}, \frac{y^{18}}{\pi \sqrt{1-y^2}}, \frac{y^{20}}{\pi \sqrt{1-y^2}}, \frac{y^{22}}{\pi \sqrt{1-y^2}}, \frac{y^{24}}{\pi \sqrt{1-y^2}}\right\}$$

$$\text{Timing}\left[\mathbf{t11} = \int_{-1}^1 \mathbf{t10} \, d\mathbf{y}\right]$$

$$\{3.85 \text{ Second}, \left\{1, \frac{1}{2}, \frac{3}{8}, \frac{5}{16}, \frac{35}{128}, \frac{63}{256}, \frac{231}{1024}, \frac{429}{2048}, \frac{6435}{32768}, \frac{12155}{65536}, \frac{46189}{262144}, \frac{88179}{524288}, \frac{676039}{4194304}\right\}\}$$

$$\mathbf{t12} = \text{Simplify}[\text{Expand}[\mathbf{t9} \cdot \mathbf{t11}]]$$

$$\begin{aligned} & \{(\lambda + 1) a(0) + 4(a(2) + 8a(4) + 27a(6) + 64a(8) + 125a(10) + 216a(12)), \\ & (\lambda + 1) a(2) + 48(a(4) + 4a(6) + 10a(8) + 20a(10) + 35a(12)), \\ & (\lambda + 1) a(4) + 24(5a(6) + 16a(8) + 35a(10) + 64a(12)), \\ & (\lambda + 1) a(6) + 16(14a(8) + 40a(10) + 81a(12)), (\lambda + 1) a(8) + 120(3a(10) + 8a(12)), \\ & (\lambda + 1) a(10) + 528a(12), (\lambda + 1) a(12)\} \end{aligned}$$

Here is the eigenvalue problem from the ODE. At this point, we have not enforced the boundary conditions so we cannot just solve this matrix for λ .

$$\mathbf{t13} = \text{Transpose}[\text{Table}[\text{Coefficient}[\mathbf{t12}, a[i]], \{i, 0, n, 2\}]]$$

$$\begin{pmatrix} \lambda + 1 & 4 & 32 & 108 & 256 & 500 & 864 \\ 0 & \lambda + 1 & 48 & 192 & 480 & 960 & 1680 \\ 0 & 0 & \lambda + 1 & 120 & 384 & 840 & 1536 \\ 0 & 0 & 0 & \lambda + 1 & 224 & 640 & 1296 \\ 0 & 0 & 0 & 0 & \lambda + 1 & 360 & 960 \\ 0 & 0 & 0 & 0 & 0 & \lambda + 1 & 528 \\ 0 & 0 & 0 & 0 & 0 & 0 & \lambda + 1 \end{pmatrix}$$

Here is the homogeneous boundary condition at $y=1$.

```
b1 = Table[1, {i, 0, n, 2}]
```

```
{1, 1, 1, 1, 1, 1, 1}
```

To enforce the boundary condition, we drop off the last equation (highest mode) and replace it with the boundary condition. Of course, this makes the eigenvalue problem singular.

```
t14 = Append[Table[t13[[i]], {i, 1,  $\frac{n}{2}$ }], b1]
```

$$\begin{pmatrix} \lambda + 1 & 4 & 32 & 108 & 256 & 500 & 864 \\ 0 & \lambda + 1 & 48 & 192 & 480 & 960 & 1680 \\ 0 & 0 & \lambda + 1 & 120 & 384 & 840 & 1536 \\ 0 & 0 & 0 & \lambda + 1 & 224 & 640 & 1296 \\ 0 & 0 & 0 & 0 & \lambda + 1 & 360 & 960 \\ 0 & 0 & 0 & 0 & 0 & \lambda + 1 & 528 \\ 1 & 1 & 1 & 1 & 1 & 1 & 1 \end{pmatrix}$$

Because eigenvalue problem is singular and we need to algebraically remove the last equation. We will solve for the highest coefficient, $a[n]$, and then use result this to eliminate $a[n]$ everywhere in the array.

Here we just take the last equation (by taking the last row, multiplying by the $a[i]$'s) and then solve for $a[n]$.

```
s1 = Solve[t14[[ $\frac{n}{2} + 1$ ]] . Table[a[i], {i, 0, n, 2}] == 0, a[n]]
```

```
{{a(12) → -a(0) - a(2) - a(4) - a(6) - a(8) - a(10)}}
```

Let's be inelegant. Remake the set of equations and then do the elimination of $a[n]$ in the rest of the system. (You can redo this another way!)

```
t15 = Table[t14[[i]] . Table[a[i], {i, 0, n, 2}], {i, 1,  $\frac{n}{2}$ }]
```

```
{(\lambda + 1) a(0) + 4 a(2) + 32 a(4) + 108 a(6) + 256 a(8) + 500 a(10) + 864 a(12),
(\lambda + 1) a(2) + 48 a(4) + 192 a(6) + 480 a(8) + 960 a(10) + 1680 a(12),
(\lambda + 1) a(4) + 120 a(6) + 384 a(8) + 840 a(10) + 1536 a(12),
(\lambda + 1) a(6) + 224 a(8) + 640 a(10) + 1296 a(12), (\lambda + 1) a(8) + 360 a(10) + 960 a(12),
(\lambda + 1) a(10) + 528 a(12)}
```

Here is the substitution for a[n] which runs off of the page.

```
t16 = Expand[t15 /. s1]
```

```
( λ a(0) - 863 a(0) - 860 a(2) - 832 a(4) - 756 a(6) - 608 a(8) - 364 a(10) - 1680 a(0) + λ a(2) - 1679 a(2) - 1
)
```

Now remake the eigenvalue matrix, which is now of lower order, but which has all finite eigenvalues.

```
t17 = Transpose[Table[
  Table[Coefficient[t16[[1, i]], a[j]], {i, 1,  $\frac{n}{2}$ }], {j, 0, n - 2, 2}]]
```

$$\begin{pmatrix} \lambda - 863 & -860 & -832 & -756 & -608 & -364 \\ -1680 & \lambda - 1679 & -1632 & -1488 & -1200 & -720 \\ -1536 & -1536 & \lambda - 1535 & -1416 & -1152 & -696 \\ -1296 & -1296 & -1296 & \lambda - 1295 & -1072 & -656 \\ -960 & -960 & -960 & -960 & \lambda - 959 & -600 \\ -528 & -528 & -528 & -528 & -528 & \lambda - 527 \end{pmatrix}$$

Here we compute the eigenvalues. Note the insertion of the - sign.

We expect them to decrease in accuracy. Thus if we want n eigenvalues, we might need n+4 or more modes in the original expansion.

Mathematica sorts these for us and does not seem to complain about computing them. I would not try to get them algebraically however!!

```
neigs = N[Eigenvalues[-t17 /. λ → 0]]
```

```
{1.4674, 21.2066, 60.7158, 126.83, 359.861, 6287.92}
```

At this point, for many problems, you might be finished. You have computed the eigenvalues.

■ Here we compute the eigenfunctions for each eigenvalue

If you need the eigenfunctions, that correspond to each of the eigenvalues, you will have to solve the matrix, `t17`, which enforces the boundary condition, for the `a[i]`'s. Of course, there cannot be a unique solution because this is an eigenvalue problem, matrix `t17` is (it better be) singular.

(Note that the eigenfunctions are probably less accurate than the eigenvalues!! Thus you will need even more modes to get the eigenfunctions correct.)

Here we again remake the equations

```
t18 = Table[t17[[i]] . Table[a[i], {i, 0, n - 2, 2}], {i, 1,  $\frac{n}{2} - 1$ }]
```

```
{(λ - 863) a(0) - 860 a(2) - 832 a(4) - 756 a(6) - 608 a(8) - 364 a(10),
 -1680 a(0) + (λ - 1679) a(2) - 1632 a(4) - 1488 a(6) - 1200 a(8) - 720 a(10),
 -1536 a(0) - 1536 a(2) + (λ - 1535) a(4) - 1416 a(6) - 1152 a(8) - 696 a(10),
 -1296 a(0) - 1296 a(2) - 1296 a(4) + (λ - 1295) a(6) - 1072 a(8) - 656 a(10),
 -960 a(0) - 960 a(2) - 960 a(4) - 960 a(6) + (λ - 959) a(8) - 600 a(10)}
```

Set them up to solve.

```
t19 = LogicalExpand[t18 == 0]
```

```
-1680 a(0) + (λ - 1679) a(2) - 1632 a(4) - 1488 a(6) - 1200 a(8) - 720 a(10) == 0 ∧
 -1536 a(0) - 1536 a(2) + (λ - 1535) a(4) - 1416 a(6) - 1152 a(8) - 696 a(10) == 0 ∧
 -1296 a(0) - 1296 a(2) - 1296 a(4) + (λ - 1295) a(6) - 1072 a(8) - 656 a(10) == 0 ∧
 -960 a(0) - 960 a(2) - 960 a(4) - 960 a(6) + (λ - 959) a(8) - 600 a(10) == 0 ∧
 (λ - 863) a(0) - 860 a(2) - 832 a(4) - 756 a(6) - 608 a(8) - 364 a(10) == 0
```

Solve them.

```
t20 = Solve[t19, Table[a[i], {i, 2, n - 2, 2}]]
```

Now we can compute the eigenfunctions for any of the eigenvalues. Here is the basic expansion. Note that I have left off the last term.

$$\text{expand2} = \sum_{\substack{i=0 \\ \Delta i=2}}^{n-2} a[i] \text{ChebyshevT}[i, y]$$

$$a(0) + (2y^2 - 1)a(2) + (8y^4 - 8y^2 + 1)a(4) + \\ (32y^6 - 48y^4 + 18y^2 - 1)a(6) + (128y^8 - 256y^6 + 160y^4 - 32y^2 + 1)a(8) + \\ (512y^{10} - 1280y^8 + 1120y^6 - 400y^4 + 50y^2 - 1)a(10)$$

To get the eigenfunction, you just substitute the $a[i]$'s and the desired λ , like this. Remember that the solution is not unique so you have an arbitrary scale factor, $a[0]$, that you can pick. For numerical problems you usually need to do some sort of normalization. We can pick the $a[0]$'s so that each eigenfunction has a value of 1 at $y=0$.

```
efunc1=(expand2/.t20)/.λ->neigs[[1]]/.a[0]->1
```

— *General::spell1* : Possible spelling error: new symbol name "efunc1" is similar to existing symbol "afunc1".

$$\{-1.05806(2y^2 - 1) + 0.0593051(8y^4 - 8y^2 + 1) - 0.00126418(32y^6 - 48y^4 + 18y^2 - 1) + \\ 0.0000142042(128y^8 - 256y^6 + 160y^4 - 32y^2 + 1) - \\ 9.85825 \times 10^{-8}(512y^{10} - 1280y^8 + 1120y^6 - 400y^4 + 50y^2 - 1) + 1\}$$

- Here are some plots of eigenfunctions from previous calculations.

```
neigs
```

```
{1.4674, 21.2066, 60.7158, 126.83, 359.861, 6287.92}
```

$\lambda = 1.4674$

```
efunc1=(expand2/.t20)/.λ->neigs[[1]]
```

$$\{-1.05806(2y^2 - 1)a(0) + \\ 0.0593051(8y^4 - 8y^2 + 1)a(0) - 0.00126418(32y^6 - 48y^4 + 18y^2 - 1)a(0) + \\ 0.0000142042(128y^8 - 256y^6 + 160y^4 - 32y^2 + 1)a(0) - \\ 9.85825 \times 10^{-8}(512y^{10} - 1280y^8 + 1120y^6 - 400y^4 + 50y^2 - 1)a(0) + a(0)\}$$


Here is where we normalize the eigenfunction to have a value of 1 at $y=0$.

```
a01=a[0]/. Solve[((efunc1/.y->0)[[1]])==1,a[0]][[1]]
```

```
0.472001
```

```
eplot1=Plot[efunc1/.a[0]->a01,{y,-1,1},  
PlotStyle->Dashing[ {.02,.02}]]
```


- Graphics -



- Graphics -

Here we compare the first analytical and numerical eigenfunctions

```
Show[aplot1,eplot1]
```



- Graphics -

$\lambda = 21.2066$

```
efunc2=(expand2/.t20)/.lambda->neigs[[2]]
```

— General::spell1 : Possible spelling error: new symbol name "efunc2" is similar to existing symbol "afunc2".


```
{1.10073 (2 y2 - 1) a(0) -
 2.78672 (8 y4 - 8 y2 + 1) a(0) + 0.773446 (32 y6 - 48 y4 + 18 y2 - 1) a(0) -
 0.0936912 (128 y8 - 256 y6 + 160 y4 - 32 y2 + 1) a(0) +
 0.0065094 (512 y10 - 1280 y8 + 1120 y6 - 400 y4 + 50 y2 - 1) a(0) + a(0)}
```

Here is where we normalize the eigenfunction to have a value of 1 at $y=0$.

```
a02=a[0]/.Solve[(efunc2/.y->0)[[1]]]==1,
a[0]][[1]]
```

-0.26588


```
eplot2=Plot[efunc2/.a[0]->a02,{y,-1,1},  
PlotStyle->Dashing[ {.02,.02}]]
```



- Graphics -

Here we compare the second analytical and numerical eigenfunctions

```
Show[aplot2,eplot2]
```



- Graphics -

$\lambda = 60.7158$

```
efunc3=(expand2/.t20)/.λ->neigs[[3]]
```

— *General::spell1* : Possible spelling error: new symbol name "efunc3" is similar to existing symbol "afunc3".


```
{1.4733 (2 y2 - 1) a(0) -
 0.679744 (8 y4 - 8 y2 + 1) a(0) - 3.38702 (32 y6 - 48 y4 + 18 y2 - 1) a(0) +
 2.04278 (128 y8 - 256 y6 + 160 y4 - 32 y2 + 1) a(0) -
 0.508787 (512 y10 - 1280 y8 + 1120 y6 - 400 y4 + 50 y2 - 1) a(0) + a(0)}
```

Here is where we normalize the eigenfunction to have a value of 1 at $y=0$.

```
a03=a[0]/.Solve[ ((efunc3/.y->0)[[1]])==1,
a[0]][[1]]
```

```
0.208963
```


```
eplot3=Plot[efunc3/.a[0]->a03,{y,-1,1},
PlotStyle->Dashing[{.02,.02}]]
```



- Graphics -

Here we compare the first analytical and numerical eigenfunctions

```
Show[aplot3,eplot3]
```



- Graphics -

For $n=12$, this one is looking a little off because the λ is inaccurate. It should be

```
N[(7 Pi/2)^2-1]
```

119.903

However if we use the computed value:

$\lambda = 126.83$

```
efunc4=(expand2/.t20)/.lambda->neigs[[4]]
```

— General::spell1 : Possible spelling error: new symbol name "efunc4" is similar to existing symbol "afunc4".

```
{1.63923 (2 y^2 - 1) a(0) +
 0.249173 (8 y^4 - 8 y^2 + 1) a(0) - 2.22354 (32 y^6 - 48 y^4 + 18 y^2 - 1) a(0) -
 2.76305 (128 y^8 - 256 y^6 + 160 y^4 - 32 y^2 + 1) a(0) +
 2.76842 (512 y^10 - 1280 y^8 + 1120 y^6 - 400 y^4 + 50 y^2 - 1) a(0) + a(0)}
```


Here is where we normalize the eigenfunction to have a value of 1 at $y=0$.

```
a04=a[0]/.Solve[ $((\text{efunc4}/.y\rightarrow 0)[[1]])==1,$   
a[0]][[1]]
```

-0.270416

This one does not match well to the analytical solution.


```
eplot4=Plot[efunc4/.a[0]->a04,{y,-1,1},  
PlotStyle->Dashing[.02,.02]]
```



- Graphics -

We see that at this point, the error is quite noticeable. This is the fourth eigenfunction. Compare this to the number of modes that were used, $= n/2+1$.


```
Show[aplot4,eplot4]
```



- Graphics -

Here are the numerical ones all together.


```
Show[eplot1,eplot2,eplot3,eplot4]
```



- Graphics -

Here are the analytical eigenfunctions.

```
Show[aplot1,aplot2,aplot3,aplot4]
```



- Graphics -

We see that this does not match real well with the same analytical modes as the mode frequency increases. The numerical ones vary in amplitude across the interval!!.

- Graphics -

- **Now you can get the eigenvalues and the eigen functions for a differential eigenvalue problem using a Chebyshev spectral method!**